
A readable and computable formalization of the
Jolteon consensus protocol

Orestis Melkonian, Mauro Jaskelioff, James Chapman
12 June 2025, TYPES @ Glasgow



Motivation

• Consensus is an integral piece of blockchain technology
• We want formally verified implementations of these protocols



Approach

1. Formally present a readable specification of the protocol
2. Provide mechanized proofs about the protocol’s properties (e.g. safety)
3. Make sure the specification is also computable

• so that we can extract executable code out of the formalization
4. Formally verifying a full implementation is too unrealistic, but...

• ...we can test that an implementation conforms to the formal model

TOOL OF CHOICE: the Agda proof assistant

Previous work in IOG

• Plutus (∼ System Fωµ) smart contract language (MPC’19, TyDe’21, FLOPS’22)
• EUTXO ledger model (WTSC’20, ISoLA’20, WTSC’24, FMBC’24, FMBC’25)
• Streamlet consensus protocol (FMBC’25)

https://iohk.io/en/research/library/

Global: states

record GlobalState : Type where
field currentTime : Time

stateMap : HonestVec LocalState
networkBuffer : List (Time × Pid × Message)
history : List Message



Global: state transition as an inductive relation

data _—→_ (s : GlobalState) : GlobalState → Type where

Deliver : ∀ {tpm}
(tpm∈ : tpm ∈ s .networkBuffer) →
───────────────────────────────
s —→ deliverMsg s tpm∈

WaitUntil : ∀ t →
∙ All (λ (t′ , _ , _) → t ≤ t′ + Δ)

(s .networkBuffer)
∙ s .currentTime < t
──────────────────────────────
s —→ record s { currentTime = t }

LocalStep : ∀ {m} ⦃ _ : Honest p ⦄ →
(p ⦂ s .currentTime ⊢ s ＠ p — m —→ ls′)
─────────────────────────────────────
s —→ broadcast m (s ＠ p ≔ ls′)

DishonestLocalStep : ∀ {m} →
∙ ¬ Honest p
∙ NoSignatureForging (m .content) s
─────────────────────────────────
s —→ broadcast (just m) s



Local View: state

record LocalState : Type where
constructor ⦅_,_,_,_,_,_,_,_,_,_,_⦆
field
r-vote : Round
r-cur : Round
qc-high : QC
tc-last : Maybe TC

inbox : Messages
db : Messages
final : Chain

...



Local View: state transition as an inductive relation

data _⦂_⊢_—_—→_ (p : Pid) (t : Time) (ls : LocalState) : Maybe Envelope → LocalState → Type where

ProposeBlock : ∀ {txs} →
let L = roundLeader (ls .r-cur)

b = mkBlockForState ls txs
m = Propose (sign L b)

in
∙ p ≡ L
────────────────────────────
p ⦂ t ⊢ ls —[ m ]—→ ls

RegisterProposal : ∀ {sb} →
let m = Propose sb

b = sb .datum
in
∀ (m∈ : m ∈ ls .inbox) →
∙ ¬ timedOut ls t
∙ sb .node ≡ roundLeader (b ∙round)
∙ ValidProposal (ls .db) b
──────────────────────────────────
p ⦂ t ⊢ ls ——→ registerProposal ls m∈

. . .



Local View: state transition as an inductive relation

data _⦂_⊢_—_—→_ (p : Pid) (t : Time) (ls : LocalState) : Maybe Envelope → LocalState → Type where

VoteBlock : ∀ {b} →
let br = (b ∙blockId , b ∙round)

m = Vote $ sign p br
L′ = nextLeader ls

in
∙ b ∙∈ ls .db
∙ ShouldVote ls b
─────────────────────────────
p ⦂ t ⊢ ls —[ L′ ∣ m ⟩—→ vote ls

Commit : ∀ {b b′ ch} →
∙ b -certified-∈- ls .db
∙ b′ -certified-∈- ls .db
∙ (b′ :: b :: ch) ∙∈ ls .db
∙ length ch > length (ls .final)
∙ b′ .round ≡ 1 + b .round
─────────────────────────────────────
p ⦂ t ⊢ ls ——→ record ls {final = b :: ch}

. . .





Mechanizing safety: closures as traces

data _—↠_ : GlobalState → GlobalState → Type where

_∎ : ∀ x →
──────
x —↠ x

_—→⟨_⟩_ : ∀ x →
∙ x —→ y ∙ y —↠ z
────────────────────
x —↠ z

Reachable : GlobalState → Type
Reachable s = s₀ —↠ s₀



Mechanizing safety: statement

safety : ∀ {s} → Reachable s →
∙ b ∈ (s ＠ p) .final
∙ b′ ∈ (s ＠ p′) .final
────────────────────────────
(b ←—∗ b′) ⊎ (b′ ←—∗ b)



Mechanizing safety: quorum intersection

uniqueCertification : ∀ {s} → Reachable s →
∙ GloballyCertified s b
∙ 1/3-HonestMajority s b′
∙ b ∙round ≡ b′ ∙round
───────────────────────
b ≡ b′



Mechanizing safety: history is complete

history-complete : ∀ {s} → Reachable s →
(s ＠ p) .db ⊆ s .history



Mechanizing safety: history is complete

history-complete (_ , refl , (_ ∎)) m∈ rewrite pLookup-replicate p initLocalState = contradict m∈
history-complete (_ , s-init , _ ⟨ st ∣ s ⟩←— tr) m∈
using Rs ← (_ , s-init , tr)
using sm ← s .stateMap
with IH ← history-complete Rs
with IH-inbox ← inbox⊆history {p = p} Rs
with st

... | WaitUntil _ _ _ = IH m∈

... | Deliver {tpm} _ rewrite receiveMsg-db {s = sm} (honestTPMessage tpm) = IH m∈

... | DishonestLocalStep _ _ = there $ IH m∈

... | LocalStep {p = p′} {ls′ = ls′} st
with p ≟ p′

... | no p≢ rewrite pLookup∘updateAt′ p p′ {const ls′} (p≢ ∘ ↑-injective) sm = ∈-++⁺ʳ _ (IH m∈)

... | yes refl rewrite pLookup∘updateAt p ⦃ hp ⦄ {const ls′} sm
with st

... | InitNoTC _ _ = IH m∈

... | InitTC _ _ = there $ IH m∈
...

... | RegisterProposal m∈inbox _ _ _ _ = go
where go : _; go with ⟫ m∈

... | ⟫ here refl = IH-inbox m∈inbox

... | ⟫ there m∈ = IH m∈





Decidability proofs as decision procedures

data Dec (P : Type) : Type where
yes : P → Dec P
no : ¬ P → Dec P

record _⁇ (P : Type) : Type where
field dec : Dec P

¿_¿ : ∀ P → ⦃ P ⁇ ⦄ → Dec P
¿ _ ¿ = dec

instance
Dec-⊥ : ⊥ ⁇
Dec-⊥ .dec = no λ()

Dec-⊤ : ⊤ ⁇
Dec-⊤ .dec = yes tt

module _ ⦃ _ : A ⁇ ⦄ ⦃ _ : B ⁇ ⦄ where instance
Dec-→ : (A → B) ⁇
Dec-→ .dec with ¿ A ¿ | ¿ B ¿
... | no ¬a | _ = yes λ a → contradict (¬a a)
... | yes a | yes b = yes λ _ → b
... | yes a | no ¬b = no λ f → ¬b (f a)

Dec-× : (A × B) ⁇
Dec-× .dec with ¿ A ¿ | ¿ B ¿
... | yes a | yes b = yes (a , b)
... | no ¬a | _ = no λ (a , _) → ¬a a
... | _ | no ¬b = no λ (_ , b) → ¬b b

Dec-⊎ : (A ⊎ B) ⁇
Dec-⊎ .dec with ¿ A ¿ | ¿ B ¿
... | yes a | _ = yes (inj₁ a)
... | _ | yes b = yes (inj₂ b)
... | no ¬a | no ¬b = no λ where (inj₁ a) → ¬a a; (inj₂ b) → ¬b b



Decidability proofs as decision procedures

instance
Dec-certified-∈ : ∀ {b ms} → (b -certified-∈- ms) ⁇
Dec-certified-∈ {b} {ms} .dec
with ¿ Any (λ qc → (qc ∙blockId ≡ b ∙blockId) × (qc ∙round ≡ b ∙round)) (allQCs ms) ¿

... | yes q = let (qc , qc∈all , (eqᵢ , eqᵣ)) = L.Mem.find q in
yes $ certified (allQCs-sound ms qc∈all) eqᵢ eqᵣ

... | no ¬q = no λ where
(certified {qc} qc∈ refl refl) →
¬q $ L.Any.map (λ x → cong proj₁ (sym x) , cong proj₂ (sym x))

(L.Any.map⁻ $ allQCs-complete ms qc∈)



Decidability proofs as decision procedures

_:RegisterProposal? : let m = _; b = sb .datum in
{_ : auto∶ m ∈ ls .inbox}
{_ : auto∶ ls .phase ≡ Receiving}
{_ : auto∶ ¬ timedOut ls t}
{_ : auto∶ sb .node ≡ roundLeader (b ∙round)}
{_ : auto∶ ValidProposal (ls .db) b}
→ s —→ _

_:RegisterProposal? {_}{x}{y}{z}{w}{q} = LocalStep $′
RegisterProposal (toWitness x) (toWitness y) (toWitness z)

(toWitness w) (toWitness q)



Example correct-by-construction traces

begin
record
{ currentTime = 10; history = [ v₂ 𝕃 ⨾ v₂ 𝔸 ⨾ p₂ ⨾ v₁ 𝔸 ⨾ v₁ 𝕃 ⨾ p₁ ]; networkBuffer = [ 10 , 𝕃 , v₂ 𝔸 ⨾ 10 , 𝕃 , v₂ 𝕃 ]
; stateMap =
[ {- 𝕃 -} ⦅ 2 , 2 , qc₁ , nothing , Receiving , _ , [] , [] , just 20 , false , false ⦆
⨾ {- 𝔸 -} ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , _ , [] , [] , nothing , false , true ⦆
⨾ {- 𝔹 -} ⦅ 0 , 1 , qc₀ , nothing , Voting , _ , _ , [] , just τ , false , false ⦆ ] }

—→⟨ 𝔹 :VoteBlock? b₁ ⟩
record
{ currentTime = 10; history = v₁ 𝔹 :: _; networkBuffer = _
; stateMap =
[ ⦅ 2 , 2 , qc₁ , nothing , Receiving , _ , [] , [] , just 20 , false , false ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , _ , [] , [] , nothing , false , true ⦆
⨾ ⦅ 1 , 1 , qc₀ , nothing , Receiving , _ , _ , [] , just τ , false , false ⦆ ] }

—→⟨ 𝔹 :RegisterProposal? ⟩
record
{ currentTime = 10; history = _ ; networkBuffer = _
; stateMap =
[ ⦅ 2 , 2 , qc₁ , nothing , Receiving , _ , [] , [] , just 20 , false , false ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , _ , [] , [] , nothing , false , true ⦆
⨾ ⦅ 1 , 1 , qc₀ , nothing , AdvancingRound , [ p₂ ⨾ p₁ ] , [] , [] , just τ , false , false ⦆ ] }

...



Example correct-by-construction traces

...
—→⟨ 𝕃 :RegisterVote? b₂ ⟩
record
{ currentTime = 13
; history = _
; networkBuffer = []
; stateMap =
[ ⦅ 2 , 2 , qc₁ , nothing , AdvancingRound , v₂ 𝔸 :: _ , v₂ 𝕃 :: _ , [] , just 20 , false , false ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , [ p₂ ⨾ p₁ ] , [] , [] , nothing , false , true ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , [ p₂ ⨾ p₁ ] , [] , [] , nothing , false , true ⦆ ] }

...



Example correct-by-construction traces

...
—→⟨ 𝕃 :RegisterVote? b₂ ⟩
record
{ currentTime = 13
; history = _
; networkBuffer = []
; stateMap =
[ ⦅ 2 , 2 , qc₁ , nothing , AdvancingRound , v₂ 𝕃 :: _ , _ , [] , just 20 , false , false ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , _ , [] , [] , nothing , false , true ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , _ , [] , [] , nothing , false , true ⦆ ] }

...



Example correct-by-construction traces

...
—→⟨ 𝕃 :Commit? [ b₂ ⨾ b₁ ] ⟩
record
{ currentTime = 13
; history = _
; networkBuffer = []
; stateMap =
[ ⦅ 2 , 3 , qc₂ , nothing , Voting , _ , _ , [ b₁ ] , nothing , false , true ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , _ , [] , [] , nothing , false , true ⦆
⨾ ⦅ 2 , 2 , qc₁ , nothing , EnteringRound , _ , [] , [] , nothing , false , true ⦆ ] }

∎





Conformance testing: trace verifier

data Action : Type where
InitTC : Action
InitNoTC : Action
ProposeBlock : List Transaction → Action

...
VoteBlock : Block → Action
Deliver : Message → Action
WaitUntil : Time → Action



Conformance testing: trace verifier

ValidTrace : List Action → GlobalState → Type
ValidTrace αs s = ∃ λ s′ → s —[ αs ]↠ s′

⟦_⟧ : ValidTrace αs s → GlobalState
⟦_⟧ = proj₁

ValidTrace-sound : (vαs : ValidTrace αs s) → s —[ αs ]↠ ⟦ vαs ⟧
ValidTrace-sound = proj₂

ValidTrace-complete : s —[ αs ]↠ s′ → ValidTrace αs s
ValidTrace-complete = -,_

instance
Dec-ValidTrace : ∀ {αs s} → ValidTrace αs s ⁇



Conclusion

We’ve demonstrated a formalization of Jolteon, which is:

• mechanized in Agda to make sure there are no mistakes;
• presented in a readable fashion;
• also computable to leverage the formal model for conformance testing.

WIP

• closing in on a liveness proof
• significantly less straightforward than safety...

• integrating trace verifier to prototype Rust implementation with nice errors, etc.

¿¿¿ Questions ¿¿¿

https://github.com/input-output-hk/formal-streamlet

https://github.com/input-output-hk/formal-jolteon

https://github.com/input-output-hk/formal-streamlet
https://github.com/input-output-hk/formal-jolteon

